Extended One-band Hubbard Model V2RDM Example

Anna O. Schouten and David A. Mazziotti

Citation:

Schouten, A. O. & Mazziotti, D. A. Entanglement Complexity in Many-body Systems from Positivity Scaling Laws. arXiv (2025) doi:10.48550/arxiv.2509.02944.

```
> with(QuantumChemistry):
  Digits := 15:
```

```
> Hubbard_one_band_PB_extended := proc(t, U, V, lat_size)
  description "Creates the one- and two-body MO integrals for the
  one-band Hubbard model with a periodic boundary";
  option `Copyright Anna O. Schouten and David A. Mazziotti`;
  local r, i, j, k, l, T1, U2, rn, U2n;
  r := lat_size;
  T1 := Array(1..r, 1..r);
  for i to r do
    for k to r do
      if (i = k + 1) or (i = k - 1) then
        T1[i,k] := -t;
      elif (i = 1) and (k = r) then
        T1[i,k] := -t;
      elif (i = r) and (k = 1) then
        T1[i,k] := -t;
      else
        T1[i,k] := 0;
      end if:
    end do:
  end do:
  U2 := Array(1..r, 1..r, 1..r, 1..r, fill = 0);
  for i to r do
    for j to r do
        if (i = j) then
            U2[i,i,j,j] := U;
        elif (i = j+1) or (i = j-1) then
            U2[i,i,j,j] := V;
            U2[j,j,i,i] := V;
        elif (i = r) and (j = 1) then
```

```
            U2[i,i,j,j] := V;
            U2[j,j,i,i] := V;
        elif (i = 1) and (j = r) then
            U2[i,i,j,j] := V;
            U2[j,j,i,i] := V;
        end if:
    end do:
  end do:
  return Matrix(ArrayTools:-Reshape(T1, [r, r])), U2;
  end proc;
```

$$Hubbard\_one\_band\_PB\_extended := \mathbf{proc}(t, U, V, lat\_size) \quad \textbf{(1)}$$

description

"Creates the one- and two-body MO integrals for the one-band Hubbard model with a periodic boundary";

...

end proc

```
> h1p, U2p := Hubbard_one_band_PB_extended(0, 1, 1, 4):
  mo[one_electron_integrals] := Matrix(h1p):
  mo[two_electron_integrals] := U2p:
  data := Variational2RDM(4, mo_integrals = mo, return_rdm =
  "rdm1_and_rdm2", conv_tol = 0.000001, conditions = "DQG");
```

$$data := \text{table}\left(\left[\left[mo\_coeff = \right.\right.\right. \quad \textbf{(2)}$$

$$\begin{bmatrix} -0.00981507 & 0.34401093 & 0.91938507 & 0.19050259 \\ 0.07699454 & 0.93131796 & -0.31222616 & -0.17097814 \\ -0.54163443 & 0.11951947 & -0.21695106 & 0.80329290 \\ 0.83702308 & -0.00429385 & -0.10088685 & 0.53776925 \end{bmatrix}, mo\_occ =$$

$$\begin{bmatrix} 1.00000001 \\ 1.00000000 \\ 0.99999999 \\ 0.99999999 \end{bmatrix}, converged = 1, group = \text{"C1"}, e\_tot = 1.99999988, rdm2$$

$$= \begin{bmatrix} 0.99999993 & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. \\ 0. & 0. & 1.99999984 & 0. \\ 0. & 0. & 0. & 0. \end{bmatrix}, rdm1 =$$

slice of 4 × 4 × 4 × 4 Array

$$\begin{bmatrix} 1.00000000 & 0. & 0. & 0. \\ 0. & 1.00000000 & 0. & 0. \\ 0. & 0. & 1.00000000 & 0. \\ 0. & 0. & 0. & 1.00000001 \end{bmatrix} \Bigg] \Bigg] \Bigg)$$

```
> data[e_tot];
```

1.99999988 **(3)**